Stability of equilibrium points

Higher-Order Scalar Difference Equations

Def. A point x* is called an equilibrium point for the difference equation

$x_{n+k} = f(x_{n+k-1}, ..., x_n)$

if x* = f(x*,...,x*)

Linear case

Consider the kth-order difference equation

$x_{n+k} + p_1 x_{n+k-1} + p_2 x_{n+k-2} + \cdots + p_k x_n = 0$

x* = 0

characteristic equation

$q^k + p_1 q^{(k-1)} + p_2 q^{(k-2)} + \cdots + p_k = 0$

Theorem. x* = 0 is asymptotically stable if and only if $|q| < 1$ for all roots of characteristic equation

Nonlinear case

$x_{n+k} = f(x_{n+k-1}, ..., x_n)$

$$p_i = \frac{\partial}{\partial u_i} f \ (x^*, ..., x^*)$$

linearized equation

$y_{n+k} = p_1 y_{n+k-1} + p_2 y_{n+k-2} + \cdots + p_k y_n$

characteristic equation corresponding to the linearized equation is

$q^k = p_1 q^{(k-1)} + p_2 q^{(k-2)} + \cdots + p_k$

Theorem (The Linearized Stability Result).

Suppose that f is continuously differentiable on an open neighborhood G from $R^{(k+1)}$ of (x*, x*, . . . , x*), where x* is an equilibrium point of the nonlinear difference equation Then the following statements are true:

(i) If all the characteristic roots of the linearized characteristic equation lie inside the unit disk in the complex plane, then the equilibrium point x* is (locally) asymptotically stable.

(ii) If at least one characteristic root of the linearized characteristic equation is outside the unit disk in the complex plane, the equilibrium point x* is unstable.

(iii) If one characteristic root is on the unit disk and all the other characteristic roots are either inside or on the unit disk, then the equilibrium point x* may be stable, unstable, or asymptotically stable.

Example

The second order Beverton.Holt model

$$x_{n+1} = \frac{r\,K\,(\alpha\,x_n + \beta\,x_{n-1})}{K + (r-1)\,x_{n-1}}$$

has been used to model populations of bottom-feeding fish, where $\alpha + \beta = 1$, r > 0, K > 0. These species have very high fertility rates and very low survivorship to adulthood. Furthermore, recruitment is
essentially unaffected by fishing. In this model, the future generation x_{n+1} depends not only on the present generation x_n but also on the previous generation x_{n-1} .

```
> f:=(u,v)->r*K*(alpha*u+beta*v)/(K+(r-1)*v);
```

$$f := (u, v) \to \frac{r\,K\,(\alpha\,u + \beta\,v)}{K + (r-1)\,v}$$

```
> eq:=x=f(x,x);
```

$$eq := x = \frac{r\,K\,(\alpha\,x + \beta\,x)}{K + (r-1)\,x}$$

```
> eqp:=solve(eq,x);
```

$$eqp := 0, \frac{K\,(-1 + r\,\alpha + r\,\beta)}{r-1}$$

```
> x1:=eqp[1];x2:=simplify(subs(beta=1-alpha,eqp[2]));
```

$$x1 := 0$$

$$x2 := K$$

```
> p1:=D[1](f)(x1,x1);
```

$$p1 := r\,\alpha$$

```
> p2:=D[2](f)(x1,x1);
```

$$p2 := r\,\beta$$

```
> lineq:=y(n+1)=p1*y(n)+p2*y(n-1);
```

$$lineq := y(n+1) = r\,\alpha\,y(n) + r\,\beta\,y(n-1)$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = r\,\alpha\,q + r\,\beta$$

```
> solve(chareq,q);
```

$$\frac{r\,\alpha}{2} + \frac{\sqrt{r^2\,\alpha^2 + 4\,r\,\beta}}{2}, \frac{r\,\alpha}{2} - \frac{\sqrt{r^2\,\alpha^2 + 4\,r\,\beta}}{2}$$

Analysing the roots x1*=0 is is locally asymptotically stable if $0 < r < 1$.

```
> p1:=D[1](f)(x2,x2);
```

$$p1 := \frac{r\,K\,\alpha}{K + K\,(r-1)}$$

```
> p1:=simplify(p1);
```

$$p1 := \alpha$$

```
> p2:=D[2](f)(x2,x2);
```

$$p2 := \frac{r\,K\,\beta}{K + K\,(r-1)} - \frac{r\,K\,(\alpha\,K + \beta\,K)\,(r-1)}{(K + K\,(r-1))^2}$$

```
> p2:=simplify(p2);
```

$$p2 := \frac{-r\,\alpha + \alpha + \beta}{r}$$

```
> p2:=subs(beta=1-alpha,p2);
```

$$p2 := \frac{-r\,\alpha + 1}{r}$$

```
> lineq:=y(n+1)=p1*y(n)+p2*y(n-1);
```

$$lineq := \mathrm{y}(n+1) = \alpha\,\mathrm{y}(n) + \frac{(-r\,\alpha + 1)\,\mathrm{y}(n-1)}{r}$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = q\,\alpha + \frac{-r\,\alpha + 1}{r}$$

```
> solve(chareq,q);
```

$$\frac{r\,\alpha + \sqrt{r^2\,\alpha^2 - 4\,r^2\,\alpha + 4\,r}}{2\,r},\ \frac{r\,\alpha - \sqrt{r^2\,\alpha^2 - 4\,r^2\,\alpha + 4\,r}}{2\,r}$$

Analysing the roots we get that the equilibrium x2* = K is locally asymptotically stable if and only if $r > 1$.

Remark. If if $r > 1$ then x2* = K is globally asymptotically stable

Numerical simulations

```
> K:=100;r:=0.8;alpha:=0.5;beta:=1-alpha;N:=100;
```

$$K := 100$$

$$r := 0.8$$

$$\alpha := 0.5$$

$$\beta := 0.5$$

$$N := 100$$

```
> f(u,v);
```

$$\frac{80.0\ (0.5\ u + 0.5\ v)}{100 - 0.2\ v}$$

```
> x[0]:=20;x[1]:=40;
```

$$x_0 := 20$$

$$x_1 := 40$$

```
> for i from 0 to N-2 do
   x[i+2]:=f(x[i+1],x[i])
end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```



```
> K:=100;r:=1.2;alpha:=0.5;beta:=1-alpha;N:=100;
```

$$K := 100$$

$$r := 1.2$$

$$\alpha := 0.5$$

$$\beta := 0.5$$

$$N := 100$$

```
> f(u,v);
```

$$\frac{120.0\ (0.5\ u + 0.5\ v)}{100 + 0.2\ v}$$

```
> x[0]:=10;x[1]:=20;
```

$$x_0 := 10$$

$$x_1 := 20$$

```
> for i from 0 to N-2 do
   x[i+2]:=f(x[i+1],x[i])
end do:
>
```

```
plot([[[n,x[n]]$n=0..N],[[n,K]$n=0..N]],style=[point,point],symbol=[circle,cross],color=[red,green]);
```




```
> restart;
>
```

Autonomous Systems of Difference equations

Linear case

$x_{n+1} = A\, x_n$

Theorem. The following statements hold:
(i) The zero solution is stable if and only if $|\lambda| \le 1$ and the eigenvalues of unit modulus are simple
(ii) The zero solution is asymptotically stable if and only if $|\lambda| < 1$ for all eigenvalues of A.

Nonlinear case

$x_{n+1} = \mathrm{f}(x_n)$

x*=f(x*) equilibrium point

linearized system

y_{n+1} = Jacobian(f) (x*) y_n

Theorem. The following statements hold:
(i) The x* is locally asymptotically stable if and only if $|\lambda| < 1$ for all eigenvalues of Jacobian(f) (x*).
(ii) The x* is unstable if there exists an eigenvalue of Jacobian(f) (x*) such that $|\lambda| > 1$.

Examples

Pielou logistic delay equation

$$x_{n+1} = \frac{\alpha\, x_n}{1 + \beta\, x_{n-1}}$$

To change this difference equation to a planar system, we let

$$\mathrm{x1}(n) = x_{n-1}, \quad \mathrm{x2}(n) = x_n$$

$$\begin{bmatrix} \mathrm{x1}(n+1) \\ \mathrm{x2}(n+1) \end{bmatrix} = \begin{bmatrix} \mathrm{x2}(n) \\ \dfrac{\alpha\, \mathrm{x2}(n)}{1 + \beta\, \mathrm{x1}(n)} \end{bmatrix}$$

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
```

```
> f:=(u,v)->vector([v,alpha*v/(1+beta*u)]);
```

$$f := (u, v) \to \left[v, \frac{\alpha\, v}{1 + \beta\, u} \right]$$

```
> f(u,v)[1];f(u,v)[2];
```

$$v$$

$$\frac{\alpha\, v}{1 + \beta\, u}$$

```
> eqp:=solve({f(u,v)[1]=u,f(u,v)[2]=v},{u,v});
```

$$eqp := \{ v = 0, u = 0 \}, \left\{ v = \frac{\alpha - 1}{\beta}, u = \frac{\alpha - 1}{\beta} \right\}$$

```
> jacobian(f(u,v),[u,v]);
```

$$\begin{bmatrix} 0 & 1 \\ -\dfrac{\alpha\, v\, \beta}{(1 + \beta\, u)^2} & \dfrac{\alpha}{1 + \beta\, u} \end{bmatrix}$$

```
> A1:=subs(u=0,v=0,jacobian(f(u,v),[u,v]));
```

$$A1 := \begin{bmatrix} 0 & 1 \\ 0 & \alpha \end{bmatrix}$$

```
> eigenvals(A1);
```

$$0, \alpha$$

```
> A2:=subs(u=(alpha-1)/beta,v=(alpha-1)/beta,jacobian(f(u,v),[u,v]));
```

$$A2 := \begin{bmatrix} 0 & 1 \\ -\frac{\alpha - 1}{\alpha} & 1 \end{bmatrix}$$

```
> eigenvals(A2);
```

$$\frac{\alpha + \sqrt{-3\,\alpha^2 + 4\,\alpha}}{2\,\alpha}, \frac{\alpha - \sqrt{-3\,\alpha^2 + 4\,\alpha}}{2\,\alpha}$$

Numerical simulation

```
> x1[0]:=10;x2[0]:=20;
```

$$x1_0 := 10$$

$$x2_0 := 20$$

```
> alpha:=0.8;beta:=1;N:=100;
```

$$\alpha := 0.8$$

$$\beta := 1$$

$$N := 100$$

```
> for i from 0 to N-1 do
  x1[i+1]:=f(x1[i],x2[i])[1];
  x2[i+1]:=f(x1[i],x2[i])[2];
end do:
> plot([[n,x1[n]]$n=0..N],style=point,symbol=circle);
```




```
> plot([[n,x2[n]]$n=0..N],style=point,symbol=circle);
```




```
> x1[0]:=10;x2[0]:=20;
```

$$x1_0 := 10$$

$$x2_0 := 20$$

```
> alpha:=3;beta:=0.3;N:=100;
```

$$\alpha := 3$$

$$\beta := 0.3$$

$$N := 100$$

```
> xx:=(alpha-1)/beta;
```

$$xx := 6.666666667$$

```
> for i from 0 to N-1 do
  x1[i+1]:=f(x1[i],x2[i])[1];
  x2[i+1]:=f(x1[i],x2[i])[2];
end do:
>
plot([[[n,x1[n]]$n=0..N],[[n,xx]$n=0..N]],style=[point,point],symbol=
[circle,cross],color=[red,green]);
```



```
> plot([[[n,x2[n]]$n=0..N],[[n,xx]$n=0..N]],style=[point,point],
symbol=[circle,cross],color=[red,green]);
```




```
>
```